

Modelování a počítačová simulace Ústav mikroelektroniky FEKT VUT v Brně			Jméno Jakub Charvot	ID 240844
			Ročník 2.	Obor MET
Spolupracoval –	Měřeno dne –	Odevzdáno dne 4.5. 2023	Hodnocení	
Název zadání Laditelná dolní propust s AD633				Č. zadání 38

Úvod

Tento projekt se zabývá analýzou analogové násobičky AD633 ve funkci laditelné dolní propusti v prostředí SPICE. Studentka verze programu PSpice [1] má bohužel omezenou funkcionalitu a také v prostředí Linux, které autor tohoto projektu používá, nefunguje bezchybně, proto byl pro vypracování použit zejména program Ngspice [2], což je jedna z open-source variant programu SPICE. Ngspice má také různé rozšiřující možnosti zápisu, které podle názoru autora této práce dělají kód o něco přehlednější, ale bohužel není zpětně kompatibilní s PSpice.

Celý zdrojový kód použitý pro simulace v tomto projektu se nachází v příloze B. Ngspice kromě výstupních souborů umožňuje také generovat data přímo do souborů typu CSV, což umožňuje následné snadné další zpracování dat, zde vykreslení do grafu v prostředí LaTeX.

1 Popis součástky


AD633 je analogová čtyřkvadrantová násobička, která se vyznačuje vysokou přesností a linearity na celém pracovním rozsahu. Při vhodném zapojení umožňuje použití v různých aplikacích: přesné analogové násobení a dělení signálů, odmocnina, násobička kmitočtu, řízená horní nebo dolní propust (která je tématem této práce) a mnohé další [3].

Součástka má 8 pinů a může pracovat v rozmezí napájecího napětí ± 8 až ± 18 V [3]. Má dvě dvojice plně diferenciálních vysokoimpedančních vstupů (X1, X2 a Y1, Y2), k tomu přičítací vstup Z. Výstup W je pak úměrný součinu napětí na vstupech s přičtenou hodnotou vstupu Z, ta je vždy přidružena k výsledku násobení (vizte vztah pro W na Obr. 1).

2 SPICE model


Model SPICE, použitý v tomto projektu byl získán přímo z webu společnosti Analog devices [4], která je výrobcem této součástky. Jedná se o poslední revizi z 15.9. 2014.

Pořadí vývodu je v modelu nastaveno stejně jako na Obr. 1. Celý kód použitého modelu je k dispozici v příloze A, se samotným modelem nebyly prováděny žádné změny.



Obr. 1: Rozložení vývodů a blokové schéma násobičky AD633. Převzato z [3].

3 Testované zapojení



Obr. 2: Schéma doporučeného zapojení napětím řízeného filtru typu DP. Převzato z [3].

V tomto zapojení (Obr. 2) je na vstup Y1 (značeno E_S) přiveden vstupní signál, který chceme filtrovat. Výsledek filtrace je pak na svorce z (značeno OUTPUT A), což není přímo výstup obvodu AD633 a je tedy zatížen vyšší výstupní impedancí, v závislosti na aplikaci může být potřeba připojit zde ještě sledovač.

Alternativně lze využít také výstup W (značeno OUTPUT B), který má nízkou výstupní impedanci a lze jej tedy více zatížit. Na nižších frekvencích se chová stejně, od frekvence f_2 už ale přestává růst jeho zeslabovací funkce a zeslabení zůstává na konstantní úrovni, znázornění ideální přenosové funkce tohoto filtru je opět na Obr. 2.

Horní mezní frekvenci f_2 filtru a také již zmíněnou frekvenci f_1 lze určit následovně:

$$f_2 = \frac{E_C}{10(2\pi RC)}$$

$$f_1 = \frac{1}{2\pi RC}$$

Z uvedených vztahů je vidět, že vlastnosti filtru jsou závislé na hodnotách součástek R a C a také na řídicím napětí E_C .

Vhodnou volbou hodnot součástek lze dosáhnout možnosti regulace mezní frekvence v námi požadovaném rozsahu pouze za pomoci napětí E_C .

4 Prováděné simulace

Jak již bylo zmíněno v úvodu, v této práci byl využíván zejména program Ngspice, ten oproti klasické syntaxi PSpice umožňuje využít sekci *.control*, ve které je možné provádět různé typy simulací, za běhu měnit hodnoty součástek nebo parametrů a také využívat příkazy podmínek a cyklů. Díky tomu je možné všechny simulace v tomto projektu spustit v rámci jednoho běhu programu.


Celý použitý zdrojový kód se nachází v příloze B, případně je k dispozici také veřejně ve formě github repozitáře [5], společně se zdrojovým kódem pro LaTeX a vygenerovanými CSV soubory.

4.1 Závislost amplitudových kmitočtových charakteristik na řídicím napětí E_C

Jak je již zmíněno v předchozí sekci, kmitočtová charakteristika filtru s AD633 je závislá na třech proměnných: Hodnotách součástek R a C a velikosti řídicího napětí E_C .

Pro účely této simulace byly zvoleny hodnoty $R=10\text{ k}\Omega$ a $C=1\text{ nF}$, které se dále neměnily a bylo pracováno pouze s velikostí řídicího napětí. Tyto hodnoty byly zvoleny po několika pokusech, poněvadž umožňují dosáhnout nejlepších výsledků při nastavování mezní frekvence filtru v akustickém pásmu frekvencí, což bylo cílem stanoveným autorem této práce.

V zadání projektu je specifikováno, že mají být simulována pouze kladná napětí, byly tedy voleny hodnoty v rozsahu od 0 do 15 V, což je velikost připojeného napájecího napětí a tedy dle katalogového listu [3] i maximální hodnota E_C . Výsledky simulace jsou zobrazeny na Obr. 3.




Obr. 3: Sít amplitudových kmitočtových charakteristik v závislosti na řídicím napětí E_C . Simulováno v Ngspice [2].

Z grafu je patrné, že pro velmi malé řídicí napětí je mezní frekvence filtru téměř nulová a i pro stejnosměrnou složku dochází k poklesu o několik dB, naopak pro řídicí napětí od zhruba 80 mV můžeme nastavovat mezní frekvenci v akustickém pásmu.


4.2 Teplotní závislost

Dále byl zkoumán vliv teploty na tento obvod. Pro všechny simulace platí, že jsou vztaženy k normované teplotě $T_{NORM} = 27^\circ\text{C}$. Nejprve byly zkoumány vlastnosti samotného obvodu s teplotně nezávislými okolními součástkami, výsledky tohoto pokusu lze vidět na Obr. 4.



Obr. 4: Teplotní závislost amplitudové kmitočtové charakteristiky pro $E_c = 100\text{ mV}$. Simulováno v Ngspice [2].

V obvodu opravdu nastaly v závislosti na teplotě drobné změny, ale jsou natolik malé, že je v grafu nelze ani okem rozlišit. V popisu SPICE modelu, který byl v této práci použit, bohužel není uvedeno, jestli je určený i pro teplotní analýzu. Ani v katalogovém listu jsem informaci o tepelné závislosti nenašel, lze tedy předpokládat, že samotný obvod bude zřejmě proti změnám teploty vcelku odolný.




Obr. 5: Teplotní závislost amplitudové kmitočtové charakteristiky pro $E_c = 100\text{ mV}$, $TC1=3000\text{ ppm}$ pro R a C. Simulováno v Ngspice [2].

Ve druhém scénáři byla součástkám R a C přidána teplotní závislost (lineární teplotní součinitel odporu/kapacity 3000 ppm), výsledky jsou zobrazeny na Obr. 5. Zde je už závislost výraznější, s rostoucí teplotou mírně klesá mezní frekvence filtru, tedy při návrhu obvodu s tímto zapojením by měl vývojář mít tuto informaci na paměti a volit součástky s dostatečně nízkým teplotním koeficientem.

4.3 Závislost horního mezního kmitočtu filtru na velikosti E_C

Z rovnice uvedené výše lze stanovit hodnotu horní mezní frekvence filtru (f_2), pro zvolené konstantní hodnoty součástek ($R=10\text{ k}\Omega$, $C=1\text{ nF}$) nám vznikne lineární závislost:

$$f_2 = \frac{E_C}{10(2\pi RC)} = E_C \cdot \frac{1}{20\pi \cdot 10 \cdot 10^3 \cdot 1 \cdot 10^{-9}} \approx 1591 \cdot E_C$$



Obr. 6: Závislost mezní frekvence na řídicím napětí. Simulováno v Ngspice [2].

Na Obr. 6 můžeme vidět, že ručně odečtené hodnoty mezní frekvence (v datech z prvního měření byly pro jednotlivá řídicí napětí vyhledány frekvence, kterým odpovídá přenos -3 dB) odpovídají uvedené rovnici, lze se tedy podle ní řídit. K odchylkám dochází až při hodnotách blížících se maximálnímu povolenému napětí.

Pro lepší citlivost a opakovatelnost měření byla nakonec vytvořena vlastní funkce pro automatickou detekci mezní frekvence z dat AC simulace ¹ a byla provedena simulace pro vyšší množství hodnot E_C . Výsledky podpořily předchozí odhad a jsou opět viditelné na Obr. 6.

¹V ngspice se tato funkce standartně nenachází nebo nebyla autorem této práce nalezena.


4.4 Vlastní experiment: zkreslení signálu

Cílem tohoto experimentu bylo zjistit, jestli po průchodu signálu filtrem dochází k jeho deformaci popř. fázovému posunu. Namísto analýzy AC tedy bylo provedena analýza TRANSIENT pro různé hodnoty frekvence vstupního harmonického signálu a následně fourierova analýza pro prvních 10 harmonických složek, z nich pak program [2] vypočítal THD ².

Simulace byla provedena pro hodnoty $R=10\text{ k}\Omega$, $C=1\text{ nF}$, $E_C = 1\text{ V}$ a testovány byly frekvence v propustném i nepropustném pásmu. Nejvyšší dosažené hodnoty THD byly v řádu tisícín procenta, tedy obvod AD633 buďto signál téměř nezkrusluje a nebo toto chování není podporováno použitým SPICE modelem [4]. Ačkoliv nebylo pozorováno zkreslení výstupního signálu, byl zde patrný fázový posun.

Ke stanovení závislosti fázového posunu byla využita data AC analýzy z první simulace, ale namísto amplitudových kmitočtových charakteristik byly na Obr. 7 zobrazeny fázové charakteristiky.

Při porovnání Obr. 3 a 7 je vidět, že v propustném pásmu k fázovému posunu téměř nedochází, po dosažení mezní frekvence už ano, ale tento signál je obecně filtrem potlačen a tedy fázový posun v této frekvenční oblasti nečiní velký problém.



Obr. 7: Sít fázových kmitočtových charakteristik v závislosti na řídícím napětí E_c . Simulováno v Ngspice [2].

²Činitel harmonického zkreslení, z anglického total harmonic distortion.

Reference

- [1] CADENCE. *OrCAD PSpice AD Student 9.1* [software]. Web update 1, build 101. 1986–1999. Dostupné také z: <https://www.orcad.com/pspice-free-trial>.
- [2] THE NGSPICE DEVELOPMENT TEAM. *Ngspice (New generation SPICE) circuit simulator* [software]. Verze 39. 2001–2022. [cit. 2023-04-23]. Dostupné z: <http://ngspice.sourceforge.net/>.
- [3] ANALOG DEVICES. *Low Cost Analog Multiplier AD633* [online]. Rev. K. 2015. [cit. 2023-04-23]. Dostupné z: <https://www.analog.com/media/en/technical-documentation/data-sheets/AD633.pdf>.
- [4] ANALOG DEVICES. *AD633 Analog Multiplier Macro Model* [online]. Rev. 2014-09-15. 1993. [cit. 2023-04-23]. Dostupné z: <https://www.analog.com/media/en/simulation-models/spice-models/ad633.cir>.
- [5] CHARVOT, Jakub. *MPS* [commit ea33ac1] [online]. GitHub. [cit. 2023-05-03]. Dostupné z: <https://github.com/charvotj/MPS>.

Přílohy

A SPICE model AD633 (ad633.lib)

```
1 * AD633 Analog Multiplier Macro Model
2 * Description: Amplifier
3 * Generic Desc: Bipolar, Multiplier, 4 Quadrant
4 * Developed by: AAG/PMI
5 * Revision History: 09/15/2014 - Input resistors nodes connection change
   (RX1A & RY1A)
6 * 08/10/2012 - Updated to new header style
7 * 1.0 (12/1993)
8 * Copyright 1993, 2012 by Analog Devices
9 *
10 * Refer to http://www.analog.com/Analog\_Root/static/techSupport/
   designTools/spiceModels/license/spice_general.html for License
   Statement. Use of this model
11 * indicates your acceptance of the terms and provisions in the License
   Statement.
12 *
13 * BEGIN Notes:
14 *
15 * Not Modeled:
16 *
17 * Parameters modeled include:
18 *
19 * END Notes
20 *
21 * Node assignments
22 *           X1
23 *           |  X2
24 *           |  |  Y1
25 *           |  |  |  Y2
26 *           |  |  |  |  VNEG
27 *           |  |  |  |  |  Z
28 *           |  |  |  |  |  |  W
29 *           |  |  |  |  |  |  |  VPOS
30 *           |  |  |  |  |  |  |  |
31 .SUBCKT AD633 1 2 3 4 5 6 7 8
32 *
33 EREF 100 0 POLY(2) 8 0 5 0 (0,0.5,0.5)
34 *
35 * X-INPUT STAGE & POLE AT 15 MHz
36 *
37 IBX1 1 0 DC 8E-7
38 IBX2 2 0 DC 8E-7
39 EOSX 10 1 POLY(1) (16,100) (5E-3,1)
40 RX1A 1 11 5E6
41 RX1B 11 2 5E6
42 *
43 GX 100 12 10 2 1E-6
44 RX 12 100 1E6
45 CX 12 100 1.061E-14
46 VX1 8 13 DC 3.05
47 DX1 12 13 DX
48 VX2 14 5 DC 3.05
49 DX2 14 12 DX
```

```

50 *
51 * COMMON-MODE GAIN NETWORK WITH ZERO AT 560 Hz
52 *
53 ECMX 15 100 11 100 10
54 RCMX1 15 16 1E6
55 CCMX 15 16 2.8421E-10
56 RCMX2 16 100 1
57 *
58 * Y-INPUT STAGE & POLE AT 15 MHz
59 *
60 IBY1 3 0 DC 8E-7
61 IBY2 4 0 DC 8E-7
62 EOSY 20 3 POLY(1) (26,100) (5E-3,1)
63 RY1A 3 21 5E6
64 RY1B 21 4 5E6
65 *
66 GY 100 22 20 4 1E-6
67 RY 22 100 1E6
68 CY 22 100 1.061E-14
69 VY1 8 23 DC 3.05
70 DY1 22 23 DX
71 VY2 24 5 DC 3.05
72 DY2 24 22 DX
73 *
74 * COMMON-MODE GAIN NETWORK WITH ZERO AT 560 Hz
75 *
76 ECMY 25 100 21 100 10
77 RCMY1 25 26 1E6
78 CCMY 25 26 2.8421E-10
79 RCMY2 26 100 1
80 *
81 * Z-INPUT STAGE & POLE AT 15 MHz
82 *
83 IBZ1 7 0 DC 8E-7
84 IBZ2 6 0 DC 8E-7
85 RZ1 7 6 10E6
86 *
87 GZ 100 32 7 6 1E-6
88 RZ2 32 100 1E6
89 CZ 32 100 1.061E-14
90 VZ1 8 33 DC 3.05
91 DZ1 32 33 DX
92 VZ2 34 5 DC 3.05
93 DZ2 34 33 DX
94 *
95 * 50-MHz MULTIPLIER CORE & SUMMER
96 *
97 GXY 100 40 POLY(2) (12,100) (22,100) (0,0,0,0,0.1E-6)
98 RXY 40 100 1E6
99 CXY 40 100 3.1831E-15
100 *
101 * OP AMP INPUT STAGE
102 *
103 VOOS 59 40 DC 5E-3
104 Q1 55 32 60 QX
105 Q2 56 59 61 QX
106 R1 8 55 3.1831E4
107 R2 60 54 3.1313E4

```

```

108 R3 8 56 3.1831E4
109 R4 61 54 3.1313E4
110 I1 54 5 1E-4
111 *
112 * GAIN STAGE & DOMINANT POLE AT 316.23 Hz
113 *
114 G1 100 62 55 56 3.141637E-5
115 R5 62 100 1.0066E8
116 C3 62 100 5E-12
117 V1 8 63 DC 4.3399
118 D1 62 63 DX
119 V2 64 5 DC 4.3399
120 D2 64 62 DX
121 *
122 * NEGATIVE ZERO AT 20 MHz
123 *
124 ENZ 65 100 62 100 1E6
125 RNZ1 65 66 1
126 FNZ 65 66 VNC -1
127 RNZ2 66 100 1E-6
128 ENC 67 0 65 66 1
129 CNZ 67 68 7.9577E-9
130 VNC 68 0 DC 0
131 *
132 * POLE AT 4 MHz
133 *
134 G2 100 69 66 100 1E-6
135 R6 69 100 1E6
136 C2 69 100 3.9789E-14
137 *
138 * OP AMP OUTPUT STAGE
139 *
140 FSY 8 5 POLY(2) VZC1 VZC2 (2.8286E-3,1,1)
141 RDC 8 5 28E3
142 GZC 100 73 72 69 11.623E-3
143 VZC1 74 100 DC 0
144 DZC1 73 74 DX
145 VZC2 100 75 DC 0
146 DZC2 75 73 DX
147 VSC1 70 72 0.695
148 DSC1 69 70 DX
149 VSC2 72 71 0.695
150 DSC2 71 69 DX
151 G01 72 8 8 69 11.623E-3
152 R01 8 72 86
153 G02 5 72 69 5 11.623E-3
154 R02 72 5 86
155 LO 72 7 1E-7
156 *
157 * MODELS USED
158 *
159 .MODEL QX NPN(BF=1E4)
160 .MODEL DX D(IS=1E-15)
161 .ENDS AD633

```

B Ngspice kód pro simulace v tomto projektu (projekt.cir)

```
1 project
2
3 ** parametry soucastek
4 .param Rx=10k
5 .param Cx=1n
6 .param Ec=100m
7 .param freq = 100 ; pro TRAN analýzu
8
9 *** NETLIST ***
10 ** napajeni
11 Vnap+ Vs+ 0 15V
12 Vnap- 0 Vs- 15V
13 Cnap1 Vs+ 0 0.1u
14 Cnap2 Vs- 0 0.1u
15 ** zbytek zapojeni
16 * Z připojeno na Y2, viz schéma
17 XAD633 X1 0 Y1 Z Vs- Z W Vs+ AD633 ; pinout modelu je: X1 X2 Y1 Y2 -Vs Z
    W +Vs
18 Rx W Z {Rx}
19 Cx Z 0 {Cx}
20 ** vstupy a vystupy
21 Vcotrol X1 0 {Ec}
22 Vsignal Y1 0 DC=0 AC=1 SIN(0 1V {freq})
23
24 .inc ad633.lib
25
26 *.AC OCT 10 10 1meg
27
28
29 .control
30 *** Bod d) ***
31 echo
32 echo "Variace Ec:"
33 echo
34 foreach val 0 10m 100m 1 10 15
35     alterparam Ec = $val
36     reset
37
38     AC OCT 10 1 1meg
39
40     * nastaveni vypisu do souboru
41     set filetype=ascii
42     set wr_singlescale
43     set wr_vecnames
44     set filename = "tex/data/d/outputEc-$val+.csv"
45     * vypis do souboru
46     wrdata $filename db(V(Z)) ph(V(Z)) ; fáze se ukládá pro zobrazení v
    bodě f)
47 end
48
49 * vykreslení grafu z této sekce
50 plot db(ac1.v(Z)) db(ac2.v(Z)) db(ac3.v(Z)) db(ac4.v(Z)) db(ac5.v(Z)) db
    (ac6.v(Z))
51
```

```

52
53
54 echo
55 echo "Variace temp:"
56 echo
57 alterparam Ec = 100m
58 reset
59
60 * AC analýza se provede pro každou z teplot v seznamu
61 foreach val -100 0 20 40 60 80 100
62     options temp = $val ; funguje stejně jako .TEMP, ale lze použít v
        sekci .control a společně s proměnnou
63     AC OCT 10 1 1meg
64
65     * nastavení výpisu do souboru
66     set filetype=ascii
67     set wr_singlescale
68     set wr_vecnames
69     set filename = "tex/data/d-teplota1/outputTemp-$val+.csv"
70     * výpis do souboru
71     wrdata $filename db(V(Z))
72 end
73 * vykreslení grafu z této sekce
74 plot db(ac7.v(Z)) db(ac8.v(Z)) db(ac9.v(Z)) db(ac10.v(Z)) db(ac11.v(Z))
        db(ac12.v(Z))
75
76
77 echo
78 echo "Variace temp podruhé:"
79 echo
80 alterparam Ec = 100m
81 reset
82 * přidání teplotního koeficientu součástkám (TC1 - lineární závislost,
        TC2 - kvadratická)
83 alter Rx TC1=0.003
84 alter Cx TC1=0.003
85
86 * AC analýza se provede pro každou z teplot v seznamu
87 foreach val -100 0 20 27 40 60 80 100
88     options temp = $val
89     AC OCT 10 1 1meg
90
91     * nastavení výpisu do souboru
92     set filetype=ascii
93     set wr_singlescale
94     set wr_vecnames
95     set filename = "tex/data/d-teplota2/outputTemp-$val+.csv"
96     * výpis do souboru
97     wrdata $filename db(V(Z))
98 end
99 * vykreslení grafu z této sekce (ac3 - 27deg nezávislé, ac16 - 27deg zá-
        vislé, ac19 - 80deg závislé)
100 plot db(ac3.v(Z)) db(ac16.v(Z)) db(ac19.v(Z))
101
102
103 *** Bod e) ***
104 echo
105 echo "Výpočet f_mez:"

```

```

106 echo
107 reset
108 setplot const ; pro nastavení globálních proměnných
109
110 * nastavení bodu, pro které budeme počítat
111 let Ec_actual = 0 ; počáteční hodnota
112 let Ec_step = 400m ; krok
113 let Ec_end = 15.2 ; maximální hodnota
114
115 * hodnoty pro výpočet -- neumím načíst z netlistu, proto natvrdo zde
116 let Rx = 10k
117 let Cx = 1n ; <<< Hardcoded !!
118
119 * nastavení vypisu do souboru
120 set filetype=ascii
121 set wr_vecnames
122 set filename = "tex/data/e/output.csv"
123
124 * samotný cyklus
125 while Ec_actual < Ec_end
126     setplot const
127     * teoreticky výpočet
128     let f_mez = Ec_actual/(20*3.141592*Rx*Cx)
129
130     *** DETEKCE ZE SIMULOVANÝCH DAT ***
131     let lastDiff = 1000
132     let actualDiff = 100
133     let lastFreq = 100
134     let actualFreq = 100
135
136     let num_points = 100 ; lze zvýšit pro lepší přesnost, výrazně ovlivňuje čas simulace
137     let ac_start_f = max(0.7*f_mez,1)
138     let ac_stop_f = 1.1*f_mez
139     echo $&num_points
140     echo $&ac_start_f
141     echo $&ac_stop_f
142
143     alterparam Ec = $&const.Ec_actual
144     reset
145     * AC analýza vždy mezi 0.7 a 1.1 násobkem teoretické hodnoty
146     AC LIN $&num_points $&ac_start_f $&ac_stop_f
147     let f_vector = frequency
148     let K_vector = db(v(Z))
149     let index = 0
150
151     **temp section for debug
152     * let tempNum = $&const.Ec_actual
153     * set tempName = $&tempNum
154     * set filena = "tex/data/temp/outputEc-$tempName+.csv"
155     * wrdata $filena db(V(Z))
156     **end temp section for debug
157
158     * v cyklu v datech z AC analýzy hledáme pokles o 3 dB
159     while const.actualDiff < const.lastDiff
160         let const.lastFreq = const.actualFreq
161         let const.actualFreq = abs(f_vector[index])
162

```

```

163     let const.lastDiff = const.actualDiff
164     let actualDiff = abs(-3-K_vector[index])
165
166     let index = index+1
167 end
168 *** KONEC DETEKCE ZE SIMULOVANYCH DAT ***
169 *** hledana frekvence je v promenne lastFreq
170
171 * výpis zjištěných údajů do souboru, v první iteraci včetně hlavičky
172 wrdata $filename const.Ec_actual const.f_mez const.lastFreq
173 unset wr_vecnames ; vypnout zahlaví
174 set appendwrite
175
176 * inkrementace cyklu
177 let const.Ec_actual = const.Ec_actual + const.Ec_step
178 end
179
180
181 *** Bod f) ***
182 *** VLASTNÍ EXPERIMENT - fourierova analýza a THD pro jednotlivé
183     frekvence vstupního signálu ***
184 foreach fval 10 100 1k 10k 100k
185     alterparam freq = $fval
186     reset
187     let timeStart = 4/$fval ; 4 periody
188     let timeEnd = 24/$fval ; 24 period
189     let step = (timeEnd-timeStart)/1000 ; 1000 vzorků
190     TRAN $&step $&timeEnd $&timeStart
191     set filename = "tex/data/f/fourier_output$fval+.txt"
192     fourier $fval V(Z) > $filename
193     plot V(Z) V(Y1) ; vykreslení výstupu z TRAN analýzy
194 end
195
196 .endc
197 .end

```

C Výsledky fourierovy analýzy ve vlastním experimentu (4.4)

C.1 $f_{signal} = 10 \text{ Hz}$

```

1 Fourier analysis for v(z):
2   No. Harmonics: 10, THD: 7.73095e-09 %, Gridsize: 200, Interpolation
3   Degree: 1
4
5
6
7
8
9
10
11
12

```

Harmonic	Frequency	Magnitude	Phase	Norm. Mag	Norm. Phase
-----	-----	-----	-----	-----	-----
0	0	-0.0037916	0	0	0
1	10000	0.893077	-27.14	1	0
2	20000	2.05193e-13	-123.68	2.29759e-13	-96.537
3	30000	6.9043e-11	-137.8	7.73091e-11	-110.66
4	40000	1.72958e-14	178.575	1.93666e-14	205.715
5	50000	3.07779e-14	10.4296	3.44628e-14	37.57
6	60000	1.71374e-14	158.134	1.91892e-14	185.274

13	7	70000	1.85429e-14	-133.75	2.07629e-14	-106.61
14	8	80000	1.16193e-14	-171.85	1.30104e-14	-144.71
15	9	90000	1.513e-14	-53.226	1.69415e-14	-26.086

C.2 $f_{signal}=100$ Hz

1	Fourier analysis for v(z):					
2	No. Harmonics: 10, THD: 8.43654e-05 %, Gridsize: 200, Interpolation Degree: 1					
3						
4	Harmonic	Frequency	Magnitude	Phase	Norm. Mag	Norm. Phase
5	-----	-----	-----	-----	-----	-----
6	0	0	-0.0037918	0	0	0
7	1	100	0.998454	-0.2922	1	0
8	2	200	2.98819e-07	-85.259	2.99281e-07	-84.967
9	3	300	2.98659e-07	-82.889	2.99122e-07	-82.597
10	4	400	2.98436e-07	-80.519	2.98898e-07	-80.227
11	5	500	2.98149e-07	-78.15	2.9861e-07	-77.858
12	6	600	2.97798e-07	-75.782	2.98259e-07	-75.49
13	7	700	2.97384e-07	-73.415	2.97844e-07	-73.122
14	8	800	2.96906e-07	-71.048	2.97366e-07	-70.756
15	9	900	2.96366e-07	-68.683	2.96824e-07	-68.391

C.3 $f_{signal}=1$ kHz

1	Fourier analysis for v(z):					
2	No. Harmonics: 10, THD: 1.81065e-11 %, Gridsize: 200, Interpolation Degree: 1					
3						
4	Harmonic	Frequency	Magnitude	Phase	Norm. Mag	Norm. Phase
5	-----	-----	-----	-----	-----	-----
6	0	0	-0.0037916	0	0	0
7	1	1000	0.997193	-2.9197	1	0
8	2	2000	4.78408e-14	-37.943	4.79755e-14	-35.023
9	3	3000	8.93961e-14	-162.02	8.96478e-14	-159.1
10	4	4000	3.61867e-14	98.4698	3.62886e-14	101.389
11	5	5000	7.93128e-14	91.0672	7.95361e-14	93.9869
12	6	6000	5.16175e-14	-21.143	5.17628e-14	-18.223
13	7	7000	7.54038e-14	-174.96	7.56161e-14	-172.04
14	8	8000	5.89699e-14	66.9209	5.91359e-14	69.8405
15	9	9000	5.37853e-14	155.005	5.39367e-14	157.924

C.4 $f_{signal}=10$ kHz

1	Fourier analysis for v(z):					
2	No. Harmonics: 10, THD: 9.92846e-05 %, Gridsize: 200, Interpolation Degree: 1					
3						
4	Harmonic	Frequency	Magnitude	Phase	Norm. Mag	Norm. Phase
5	-----	-----	-----	-----	-----	-----
6	0	0	-0.0037918	0	0	0
7	1	10	0.998456	-0.029268	1	0
8	2	20	3.52322e-07	-84.865	3.52866e-07	-84.835
9	3	30	3.52028e-07	-82.298	3.52572e-07	-82.268
10	4	40	3.51617e-07	-79.732	3.52161e-07	-79.703
11	5	50	3.5109e-07	-77.168	3.51633e-07	-77.138
12	6	60	3.50447e-07	-74.605	3.50988e-07	-74.576
13	7	70	3.49688e-07	-72.044	3.50228e-07	-72.015

14	8	80	3.48814e-07	-69.486	3.49353e-07	-69.457
15	9	90	3.47827e-07	-66.931	3.48364e-07	-66.902

C.5 $f_{signal} = 100$ kHz

1	Fourier analysis for v(z):					
2	No. Harmonics: 10, THD: 0.00229586 %, Gridsize: 200, Interpolation					
3	Degree: 1					
4	Harmonic	Frequency	Magnitude	Phase	Norm. Mag	Norm. Phase
5	-----	-----	-----	-----	-----	-----
6	0	0	-0.0037924	0	0	0
7	1	100000	0.195618	-84.544	1	0
8	2	200000	1.60114e-06	-84.241	8.18503e-06	0.303055
9	3	300000	1.59298e-06	-81.794	8.14334e-06	2.74953
10	4	400000	1.59634e-06	-78.485	8.16049e-06	6.05849
11	5	500000	1.59274e-06	-75.611	8.1421e-06	8.93337
12	6	600000	1.58836e-06	-72.738	8.11972e-06	11.8061
13	7	700000	1.58319e-06	-69.868	8.09331e-06	14.6757
14	8	800000	1.57725e-06	-67.002	8.06292e-06	17.5419
15	9	900000	1.57053e-06	-64.14	8.02858e-06	20.4041